

Als einziges Serienlaufwerk besitzt die Floppy 2000 die Kombination aus 360KB Schreibdichte der XF551 und dem High-Speed einer aufgerüsteten 1050 mit Speedy.

Aber in der Floppy 2000 stecken wesentlich mehr Möglichkeiten mit denen sich mehr aus der Floppy herausholen lässt. Es soll sowohl der versierte Programmierer als auch der einfache User angesprochen werden.

Die High-Speed der Floppy 2000

Wie du aus dem Handbuch der FLOPPY 2000 weißt, handelt es sich um ein komplettes Prozessorsystem mit RAM, ROM, Ein-/Ausgabeperipherie und der Controllerlogik. Mittels der RAM-Unterstützung erreicht das Laufwerk die hohe Arbeitsgeschwindigkeit.

High-Speed

Um nun in den Genuss der hohen Übertragungsrate zu kommen muss eine High-Speed-SIO Routine her.

Im Normalfall arbeitet die SIO (Serial Input/Output Routine) des XL/XE Betriebssystemes mit einer Übertragungsrate von 19200 Baud. Durch den Trackbuffer der FLOPPY 2000 werden auch hier schon die Daten zum Computer schneller übertragen als bei einer normalen 1050.

Installation

So richtig zum laufen kommt die FLOPPY 2000 aber erst wenn eine High-Speed-SIO im Computer installiert wird.

Diese High-Speed-SIO findet man z.B. in der Fast-Version des Bibo-DOS, im Copy2000 oder in der ROM-Software.

Den Unterschied bemerkst du bei der Benutzung von DOS 2.5 oder der Bibo-DOS Normal Version. Schreibst du ein eigenes Programm und willst die High-Speed nutzen, so brauchst du die HS-SIO-Routine.

Für den Assembler-Programmierer ist es ein leichtes die HS-SIO zu installieren. Diese braucht nicht erst geschrieben zu werden, Sie befindet sich schon im ROM der Floppy 2000 und muss nur „abgeholt“ werden.

Durch die Relokater-Tabelle im Laufwerk wird die HS-SIO an jede gewünschte Adresslage im Speicher angepasst. Die Arbeit des Relokieren übernimmt der Prozessor des Laufwerks.

Diese kleine Assembler-Listing lädt die HS-SIO in den Computer. Wird in Zeile 40 eine andere Adresse angegeben, steht die HS-SIO an dieser Adresse zur Verfügung. In Targetadresse wird die Adresse eingetragen, an welche die HS-SIO geladen werden soll. In Relokieradresse wird die Adresse eingetragen an welche die HS-SIO später laufen soll.

Dies ist dann von Nutzen wenn die HS-SIO zu einem späteren Zeitpunkt an eine andere Adresse verschoben werden soll. Angesprungen werden darf die HS-SIO nur an der Relokieradresse.

Wird nun bei allen Diskettenoperationen nicht die \$E459, sondern die HS-SIO angesprungen, werden alle Disketten-Operationen in High-Speed durchgeführt.

Beachte dass die HS-SIO nur zum Übertragen von Daten, nicht zum Übertragen von Kommandos benutzt werden darf. Eine Funktionseinschränkung gibt es hier zwar nicht, aber bei evtl. auftretenden Fehlern kann es durchaus sein dass Computer und Laufwerk bis zu einer Minute aufeinander „warten“.

```

00010      .LI OFF
00020      .OR $4000
00030 ;
00040 ADR      = $5000      ; hier die gewünschte Adresse für HS-SIO
00050 ;
00060      LDA #$31      ; BUS ID
00070      STA $0300     ; DSKDEVICE
00080      LDA #1        ; Laufwerksnummer für nachfolgendes Kommando
00090      STA $0301     ; DSKUNIT
00100      LDA #$68     ; Kommando $68, Länge der HS-SIO ermitteln
00110      STA $0302     ; DSKCMD
00120      LDA #$40     ; Status für Daten lesen
00130      STA $0303     ; DSKSTATUS
00140      LDA #8        ; Low-Byte des Daten-Buffers
00150      STA $0304     ; DSKBUFFER-low
00160      STA $0306     ; DSKTIMEOUT-low, Sekunden bis Timeout .ca 8
00170      LDA #3        ; High-Byte des Daten-Buffers
00180      STA $0305     ; DSKBUFFER-high
00190      LDA #2        ; Anzahl der zu lesenden Bytes, hier 2
00200      STA $0308     ; DSKBYTCNT-low
00210      LDA #0        ; ...
00220      STA $0309     ; DSKBYTCNT-high
00230      JSR $E459     ; Einsprungadresse für XL/XE-SIO
00240      BMI ERROR     ; Wenn Fehler
00250      INC $0302     ; Jetzt Kommando $69, HS-SIO senden
00260      LDA #ADR      ; Low-Byte der Adresse für HS-SIO im Computer
00270      STA $0304     ; Targetadresse für HS-SIO, wohin laden? low-Byte
00280      STA $030A     ; Relokieradresse HS-SIO, wo lauffähig? low-Byte
00290      LDA /ADR     ; High-Byte der Adresse für HS-SIO im Computer
00300      STA $0305     ; Targetadresse für HS-SIO high-Byte
00310      STA $030B     ; Relokieradresse für HS-SIO high-Byte
00320      LDA #$40     ; Status für Daten lesen
00330      STA $0303     ; DSKSTATUS
00340      JSR $E459     ; Einsprungadresse für XL/XE-SIO
00350      BMI ERROR     ; Wenn Fehler
00360      CLC           ; Carry gelöscht, alles klar
00370      RTS
00380 ERROR   SEC           ; Carry gesetzt, FEHLER!
00390      RTS

```

Die HS-SIO funktioniert übrigens genauso bei der 1050 mit Speedy. Das Übertragungsbyte der Laufwerke wird von der HS-SIO eigenständig ermittelt, so dass es auch keine Probleme mit normalen Laufwerken gibt.

Zum Ermitteln der Übertragungsrate existiert das Kommando \$3F. Wird bei einem normalen Laufwerk das Kommando \$3F gesendet, wird von der SIO ein Fehler zurückgesendet. Floppy 2000 und Speedy 1050 übertragen hier \$0A bzw. \$09. Je höher der Wert, desto höher die Übertragungsrate.

Aber nicht wundern, die Speedy 1050 ist nicht schneller als die Floppy 2000. Das Byte \$0A kommt durch die höhere Taktrate zustande (Prozessortakt von 1,04 MHz). Ein normales Laufwerk arbeitet mit \$10, dieser Wert wird von der SIO (im XL/XE OS) fest benutzt.

Betriebssystem SIO im XL/XE

SIO (Serial Input/Output) ist eines der Herzstücke des Betriebssystems (nachfolgenden OS genannt). Die SIO erledigt alle Operationen für Diskettenstation und Datasette. Erreicht wird die SIO durch einen Sprung auf den Vektor \$E459.

Wie wir wissen befindet sich ab Adresse \$E450 eine Sprungtabelle, welche wiederum einen Sprung an die erforderliche Betriebssystemadresse enthält. Man bewahre sich diesen Zwischensprung zu sparen und direkt in das OS zu „jumpen“. Da das OS im Laufe der Jahre einige Veränderungen erfahren hat, ist der direkte Sprung nicht immer von Erfolg gekrönt. Die Garantie über den Vektor hat man jedoch immer, egal welche Version.

Natürlich kann man nicht einfach in das OS „jumpen“ und anschliessend erwarten, dass von hier aus gewusst wird, was zu tun ist.

Die SIO arbeitet die OS-Routinen unter Berücksichtigung einiger vom Programmierer selbst bestimmten Variablen ab. Diese Variablen befinden sich ab Adresse \$300 und haben folgende Bedeutung:

| | |
|-------------------|---|
| \$0300 DSKDEVICE | Erkennungsbyte für Disk (\$31) oder Cass (\$61) |
| \$0301 DSKUNIT | Hier wird die Laufwerksnummer eingetragen |
| \$0302 DSKCMD | Auszuführendes Kommando, siehe unten |
| \$0303 DSKSTATUS | Zurückgemeldeter Status der I/O-Operationen |
| \$0304 DSKBUFFER | Anfangsadresse des Datenbuffers - low-Byte |
| \$0305 DSKBUFFER | Anfangsadresse des Datenbuffers - high-Byte |
| \$0306 DKSTIMEOUT | Timeout in Sekunden, dann Fehlermeldung - low-Byte |
| \$0307 DKSTIMEOUT | Timeout in Sekunden, dann Fehlermeldung - high-Byte |
| \$0308 DSKBYTCNT | Zahl der im Buffer stehenden Bytes - low-Byte |
| \$0309 DSKBYTCNT | Zahl der im Buffer stehenden Bytes - high-Byte |
| \$030A DSKAUX1 | Hilfsbyte 1 zur Diskverarbeitung |
| \$030A DSKAUX2 | Hilfsbyte 2 zur Diskverarbeitung |

Dies sind alle in Frage kommenden Bytes, welche je nach Kommando, vor dem SIO-Aufruf bearbeitet werden müssen.

Eine besondere Bedeutung hat hier DSKCMD (\$0302). Hier wird das Kommando eingetragen. Gültige Kommandos sind:

| | | |
|----------------|------|--|
| GET Sector | \$52 | Sektor lesen |
| PUT Sector | \$50 | Sektor schreiben |
| WRITE Sector | \$57 | Sektor schreiben wie \$50, jedoch mit Verify |
| STATUS Request | \$53 | Statusabfrage der Diskettenstation |
| FORMAT Disk | \$21 | Diskette formatieren SD |
| FORMAT Disk | \$22 | Diskette formatieren MD |

Dies sind die Kommandos welche jedes Laufwerk für XL/XE beherrscht. Hierzu ein kleines Assembler-Listing.

Um den Sektor zu lesen muss WRITESEC mit dem Kommando \$52 sowie einer Adresse für die Daten (z.B. \$5000) ausgestattet werden, dann wird Sektor 1 wieder gelesen (nicht formatieren sonst ist dieser leer).

Als kleinen Gag solltest du mal das Floppy 2000 Kommando \$20 zur Formatierung benutzen, ohne den Sektor 1 zu beschreiben. Wenn die Diskette anschliessend booten soll erscheint nicht das gewöhnliche „BOOT ERROR“.

```

00010      .LI OFF
00020      .OR $4000
00030 *
00040 DSKDEVICE =      $0300
00050 DSKUNIT   =      $0301
00060 DSKCMD    =      $0302
00070 DSKSTATUS =      $0303
00075 DSKBUFFER =      $0304
00080 DSKTIMEOUT =      $0306
00090 DSKBYTCNT =      $0308
00100 DSKAUX1   =      $030A
00110 DSKAUX2   =      $030B
00120 *
00130 SIO       =      $E459
00140 WRBUFF   =      $E000      ; Zeichensatz als Datenquelle
00150 REBUFF    =      $5000
00160 *
00170 FORMAT    LDA $31
00180           STA DSKDEVICE
00190           LDA #1
00200           STA DSKUNIT
00210           LDA $22      ; Medium-Format
00220           STA DSKCMD
00230           LDA #0
00240           STA DSKBUFFER
00250           LDA $50
00260           STA DSKBUFFER+1
00270 ;
00275 ; Für das Format-Kommando reichen Unit-Nr., Device, Buffer mit 128
00280 ; Byte, bzw. 256 Byte bei Double/Quad-Density und das Kommando.
00285 ; Anschliessend erster Sprung nach SIO.
00290 ;
00300           JSR SIO
00310           BMI ERROR
00320 ;
00325 ; Unter ERROR sollte eine Abfang-Routine stehen, hierin wird bei Fehler
00330 ; verzweigt. ACHTUNG: Disk wird hier FORMATIERT.
00335 ;
00340 WRITESEC   LDA $50      ; Scheiben ohne Verify
00350           STA DSKCMD
00360           LDA $80      ; STATUS für Scheiben
00370           STA DSKSTATUS
00380           LDA #WRBUFF  ; Low-Byte der Daten-Adresse
00390           STA DSKBUFFER
00400           LDA /WRBUFFER ; High-Byte der Daten-Adresse
00410           STA DSKBUFFER+1
00420           LDA #7      ; Timeout 7 Sekunden
00430           STA DSKTIMEOUT
00440           LDA $80      ; 128 Byte schreiben
00450           STA DSKBYTCNT
00460           LDA #0
00470           STA DSKBYTCNT+1
00480           LDA #1      ; Bootsektor beschreiben
00490           STA DSKAUX1
00500           LDA #0      ; Sektornummer high (hier 0)
00510           STA DSKAUX2
00520           JSR SIO
00530           BMI ERROR  ; wie üblich
00540           CLC
00550           RTS
00560 ERROR     SEC
00570           RTS

```

Im folgenden SIO-Programm wird auch gezeigt dass man die Floppy 2000 seinen eigenen Bedürfnissen anpassen, sprich konfigurieren kann. Wie du schon erfahren hast muss dafür die SIO vorbereitet werden. Da es sich beim konfigurieren um einen reinen Lese-Befehl handelt, ist es nicht ganz so aufwendig.

```

00010      .LI OFF
00020      .OR $4000
00030 *
00040 DSKDEVICE   =      $0300
00050 DSKUNIT    =      $0301
00060 DSKCMD     =      $0302
00070 DSKSTATUS  =      $0303
00080 DSKTIMEOUT =      $0306
00090 DSKAUX1   =      $030A
00100 *
00110 SIO        =      $E459
00115 *
00120 Start      LDA $31           ; Disklaufwerk
00130            STA DSKDEVICE
00140            LDA #1            ; Laufwerksnummer
00150            STA DSKUNIT
00160            LDA #$4B         ; Kommando $4B, Erklärung unten
00170            STA DSKCMD
00180            LDA #0           ; keine Daten senden oder Empfangen
00190            STA DSKSTATUS
00200            LDA #7           ; Timeout 7 Sekunden
00210            STA DSKTIMEOUT
00220            LDA #3           ; Hier kann jeder Wert stehen, siehe Konfiguration
00230            STA DSKAUX1
00240            JSR SIO
00250            BMI ERROR        ; immer dran denken!
00260            CLC
00270            RTS
00280 ERROR      SEC
00290            RTS

```

In Zeile 160 steht hier das Kommando \$4B, welches nur auf Floppy 2000-II und Speedy 1050 angewandt werden kann. Die Atari-Laufwerke 1050 und XF551 stehen hier aussen vor. Wenn du ein solches Kommando in eigene Programme einbaust die mit allen Laufwerken funktionieren sollen, dann nur bei o.g.

Kommando Slow/Fast \$4B

Hierzu sollte vorher das Kommando \$3F gesendet werden, 1050 und XF551 geben hier eine Fehlermeldung zurück. Unter BMI ERROR könnte dann der weitere Programmablauf einer „normalen“ Floppy erfolgen. Das Kommando \$4B dient zur „Slow/Fast“ Konfiguration des Laufwerks. Heute wird sehr viel Software angeboten die die Ultra-Speed der Floppy 2000 oder Speedy nutzen. Darüber hinaus wird auch gelegentlich ein Kopierschutz aufgebracht.

Da aber nach Art des Schutzes (z.B. doppelter Sektor mit Zeitmessung) eine Floppy 2000 hier anders reagieren würde, die Schutzabfrage also auch beim Original negativ ausfallen würde, kann mit diesem Befehl dafür gesorgt werden, dass z.B. der nächste Sektor „Slow“ gelesen wird. Danach kann wieder mit Ultra-Speed gearbeitet werden.

Der Wert der hierbei nach DSKAUX1 (Zeile220,230) geschrieben wird, kann dabei je nach Bedarf angepasst werden. Der Wert ist Binär zu betrachten, wobei die Bedeutung der gesetzten Bits (1) folgende ist:

```

Bit 0 :      Read Sector Slow
Bit 1 :      Write Sector Slow
Bit 3 :      Kommando $57, Verify AUS
Bit 4 :      unbenutzt
Bit 5 :      NUR Speedy 1050 : Cache AUS *
Bit 6 :      einen Track Slow, nach Trackwechsel 0
Bit 7 :      eine Diskette Slow, nach Disk-Wechsel 0

```

Die oben aufgeführten Funktionen werden ausgeführt bei gesetztem Bit (1). Bei setzen der Bits 0 und 1 wird solange SLOW gelesen/geschrieben bis das Bit wieder gesetzt wird. Die Bits 0 bis 3 können auch über das Setup der Floppy 2000 vom User gesetzt/gelöscht werden. Das Bit 5 kommt nur für die Speedy 1050 in Frage da es bei der Einführung der Quad-Density vom OS der Floppy 2000 immer auf 1 gesetzt wird *.

Bei setzen der Bits 0, 1, 2, 6 und 7 können diese mittels Kommando wieder gelöscht werden. Wird jedoch das Laufwerk vollständig „SLOW“ geschaltet (Bit 3), kann nur noch über Aus- und wieder Einschalten die alte Konfiguration wieder hergestellt werden.

Im SLOW-Mode wird auch die Erweiterte Kommandotabelle, damit auch Kommando \$4B, nicht mehr erkannt. Da also dieses Byte nicht mehr gelesen werden kann müssen immer alle Bits richtig gesetzt sein.

* Das setzen von Bit 5 auf 1 in der Floppy 2000 geschieht automatisch. Das schreiben des Cache auf Quad-Density-Disketten machte Ärger, da die Cache-Steuerung Timingabhängig erfolgt. Durch das neue „Sideselect“ (Seitenwahl bei Quad-Density), welches von den Cache-Routinen nicht erkannt wird, wurde teilweise der Cache auf der falschen Diskettenseite geschrieben. Aus diesem Grund musste dieser Cache-Buffer entfallen.

Disk Status §53

Kommst du mit der Statusabfrage nicht ganz zurecht, sieh dir nochmals das Kommando §3F an. Da die SIO des XL/XE hierfür schon vorbereitet ist, wird DSKBUFFER auf DSKSTATUS gesetzt. Dies heisst, es darf (kann) kein Buffer für die 4 Byte angegeben werden. Die Bytes stehen danach ab DSKSTATUS (§0303) zur Verfügung. Von Bedeutung sind hierbei die Bytes 1 und 2. Byte 3 enthält den Timeout-Wert beim formatieren (normal §E0) , Byte 4 ist unbenutzt (immer 0).

Im Byte 1 steht der Laufwerksstatus:

| | | |
|-------|-----------------------|---|
| Bit 0 | : Command Frame Error | - ist gesetzt nach einem ungültigen Lesekommando |
| Bit 1 | : Checksum Error | - gesetzt nach lesen eines defekten Sektors |
| Bit 2 | : Operation Error | - gesetzt nach erfolglosem Schreibversuch (PUT) |
| Bit 3 | : Write protect | - gesetzt bei schreibgeschützter Diskette |
| Bit 4 | : Motor on | - gesetzt bei Diskmotor EIN |
| Bit 5 | : Double Density | - gesetzt bei Double Density (Erweiterte Floppys) |
| Bit 6 | : unbenutzt | |
| Bit 7 | : Medium Density | - gesetzt bei Medium Density |

Im Byte 2 stehen dir jede Menge Informationen bezüglich Track und Sektorbeschaffenheit zur Verfügung. Dieses Statusbyte ist z.B. beim Programmieren von Backup-Programmen (Kopierschutzknacker) von herausragender Bedeutung.

Im Byte 2 steht der Controllerstatus:

| | |
|-------|--------------------|
| Bit 0 | : Busy |
| Bit 1 | : DRQ |
| Bit 2 | : Lost Data |
| Bit 3 | : CRC Error |
| Bit 4 | : Record not found |
| Bit 5 | : Record Type |
| Bit 6 | : Write Protect |
| Bit 7 | : Not Ready |

Die Möglichkeit den Controller WD2793 ganz nach seinen Bedürfnissen zu Benutzen, verleiht der Floppy 2000 seine ganz immensen Fähigkeiten. Aber die Floppy 2000 hat natürlich noch mehr zu bieten, z.B. den Programmspeicher für eigene Floppy-Routinen. Weitere Merkmale sind das Einfügen eigener Floppy-Kommandos, löschen vorhandener Kommandos und ersetzen.

Die speziellen Speedy 1050 / Floppy 2000 Kommandos:

| | | |
|------------------|-----|--|
| FORMAT Disk Auto | §20 | Automatisches formatieren (sofortige Rückmeldung) |
| SIO-Speed | §3F | High-Speed-SIO senden (Speedy §09, F2000 §0A) |
| Kommandotabelle | §41 | Kommandotabelle verlängern oder verkürzen |
| Drive/Display | §44 | Das Byte für Drive/Display wird gesetzt |
| Slow/Fast | §4B | Konfigurieren der Slow/Fast-Kommandos |
| Sprungbefehl O | §4C | Springe zur Speicherstelle in AUX1/AUX2 ohne Rückmeldung |
| Sprungbefehl M | §4D | Springe zur Speicherstelle in AUX1/AUX2 mit Rückmeldung |
| Drive-Konfig. | §4E | Die Laufwerkskonfiguration wird ausgelesen |
| Drive-Konfig. | §4F | Die Laufwerkskonfiguration wird gesetzt |
| Track write | §60 | Track lesen |
| Track read | §61 | Track schreiben |
| Länge SIO | §68 | Länge der SIO-Routine ermitteln |
| Länge SIO | §69 | Länge der SIO-Routine setzen |

Siehe hierzu den Text „Das Speedy 1050 System“